- Supporting Information. Elahi, R., Edmunds, P.J., Gates, R.D., Kuffner, I.B., Barnes, B.B.,
- ² Chollett, I., Courtney, T.A., Guest, J.R., Lenz, E.A., Toth, L.T., Viehman, T.S., Williams, I.D.
- ³ 2021. Coral reef oases are consistently associated with high light attenuation. Ecological
- 4 Applications.

5 Appendix S2.

- 6 Any use of trade, firm, or product names is for descriptive purposes only and does not
- ⁷ imply endorsement by the U.S. Government.

8 Section S1. Statistical Model

We were interested in understanding the probability of occurrence of coral reef oases, and in particular, how environmental covariates mediated their probability of occurrence. As stated in the *Methods*, we defined an oasis as a reef site that exhibited higher coral cover relative to reef sites within a defined area. These areas were divided into 2.5 arcminute grid cells (\sim 21.2 km²), and individual sites (< 100 m²) were sampled within these grid cells. The probability of occurrence can be separated into the true probability of occurrence of an oasis (ψ), and the probability of detecting an oasis (p). The true probability of occurrence (ψ) was modeled using a deterministic equation with the pertinent environmental covariates, and the detection probability (p) was modeled using a binomial distribution with the number of trials (sampled reefs) and successes (oases). Our modeling approach is based on species occupancy models (Mackenzie et al. 2002), which estimate species occupancy (i.e., occurrence) when detection probabilities are less than one.

Our data set was the number of oases (y_{ij}) observed in a given cell i nested within subregion j, given n_{ij} sampling occasions. We wished to predict the true probability of occurrence of an oasis for cell i. We defined the unobserved, true state of cell i as $z_{ij} = 1$ if it had an oasis, and $z_{ij} = 0$ if it did not. Then we modeled the data y_{ij} , the number of times we observed an oasis given n_{ij} sampling occasions as:

$$y_{ij} \sim \begin{cases} 0, & \text{if } z_{ij} = 0 \\ \text{Binomial}(n_{ij}, p_j), & \text{if } z_{ij} = 1 \end{cases}$$

which states that we will never detect an oasis site if the cell does not have one, but if the cell does have an oasis, we will detect it with probability p_j , estimated as a group-level intercept for sub-region j, designated as $h(\eta_j)$ below.

Next, we modeled the process governing the true state z_{ij} :

$$z_{ij} \sim \text{Bernoulli}(\psi_{ij})$$

We used a Bernoulli distribution because the random variable z_{ij} can take on values of 0 or 1. The frequency of these values is determined by the true probability of occurrence, ψ_{ij} . 31 We modeled ψ_{ij} using a deterministic model, $g(\alpha_j, \beta, x_{ij})$, where α_j represented an intercept 32 for sub-region j, β represented a vector of coefficients, and x_{ij} represented a vector of 33 the measured covariates for $cell_{ij}$. The covariates were assumed to be measured without 34 error and thus were not treated as random variables in our model. We used an inverse 35 logit function because it returns continuous values from 0 to 1. Finally, we calculated the 36 posterior probability of our random variables conditional on our data using the following 37 Bayesian hierarchical model:

$$[z, \boldsymbol{\beta}, \boldsymbol{\alpha}, \boldsymbol{\eta}, \mu_{\alpha}, \sigma_{\alpha}, \mu_{\eta}, \sigma_{\eta} | \boldsymbol{n}, \boldsymbol{y}] \propto \prod_{i=1}^{890} \prod_{j=1}^{32} [y_{ij} | n_{ij}, h(\eta_{ij}) z_{ij}] \quad \text{(detection model)}$$

$$\times [z_{ij} | g(\alpha_j, \boldsymbol{\beta}, \boldsymbol{x}_{ij})] \quad \text{(occurrence model)}$$

$$\times [\alpha_j | \mu_{\alpha}, \sigma_{\alpha}] \quad \text{(occurrence hyperprior)}$$

$$\times [\eta_j | \mu_{\eta}, \sigma_{\eta}] \quad \text{(detection hyperprior)}$$

$$\times [\boldsymbol{\beta}] [\mu_{\alpha}] [\sigma_{\alpha}] [\mu_{\eta}] [\sigma_{\eta}] \quad \text{(priors)}$$

$$g(\alpha_j, \boldsymbol{\beta}, x_{ij}) = \text{invlogit}(\alpha_j + \boldsymbol{\beta} x_{ij})$$


$$h(\eta_{ij}) = \text{invlogit}(\eta_j)$$

with the following priors:

```
eta \sim \text{Normal}(0,1)
lpha_j \sim \text{Normal}(\mu_{lpha}, \sigma_{lpha})
\mu_{lpha} \sim \text{Normal}(-1,1)
\sigma_{lpha} \sim \text{Exponential}(1)
\eta_j \sim \text{Normal}(\mu_{\eta}, \sigma_{\eta})
\mu_{\eta} \sim \text{Normal}(-1,1)
\sigma_{\eta} \sim \text{Exponential}(1)
```

We used weakly regularizing priors for the slope coefficients (β), noting that the environmental covariates were standardized to have a mean of 0 and standard deviation of 1 41 (Fig. S1A). We also chose weakly regularizing priors for the hyperpriors μ_{α} (Fig. S1B) and 42 σ_{α} (Fig. S1C) so that their resulting group-level intercepts (α_i) for the true probability of 43 occurrence peaked between 0 and 0.3, and then declined steadily towards 1 (Fig. S2A). We 44 chose to put more weight on probabilities less than 0.5 because oases are, by definition, rare 45 occurrences. The same priors were used for hyperpriors μ_{η} and σ_{η} , for the same reasons. 46 We visualized our prior predictive distributions to ensure that the resulting relationships 47 between the true probability of occurrence and a standardized coefficient were reasonable 48 (Fig. S2B).

50 Section S2. Supporting Figures



- $_{53}$ Figure S1. Predictive distributions based on 10000 simulated draws for the priors from the
- ⁵⁴ hierarchical model. Parameters are on the logit scale.




Figure S2. Predictive distributions based on 10000 simulated draws for the priors from the hierarchical model. In (A), the probability of oasis occurrence is back-transformed from the logit scale. In (B), 25 randomly selected relationships between the true probability of occurrence and a standardized covariate derived from the prior predictive distributions.

60 Section S3. JAGS model

```
model{
  # priors for occurrence model
  for(i in 1:nX){
    beta[i] ~ dnorm(0, 1)
  }
  # hyper-priors for occurrence model
 mu.alpha ~ dnorm(-1, 1)
  sigma.alpha ~ dexp(1)
  tau.alpha <- 1/sigma.alpha^2</pre>
  for(j in 1:y.n.sites){
    alpha[j] ~ dnorm(mu.alpha, tau.alpha)
  }
  # hyper-priors for detection model
  mu.det ~ dnorm(-1, 1)
  sigma.det ~ dexp(1)
  tau.det <- 1 / sigma.det^2</pre>
  for(j in 1:y.n.sites){
    eta[j] ~ dnorm(mu.det, sigma.det)
  }
  # likelihood
  for(i in 1:N){
```

```
# occurrence model
  logit(psi[i]) <- alpha[y.group[i]] + inprod(beta[], X[i, ] )</pre>
  z[i] ~ dbern(psi[i])
  # detection model
  logit(p[i]) <- eta[y.group[i]]</pre>
  mu.p[i] <- z[i] * p[i]
 y[i] ~ dbin(mu.p[i], n[i])
  # simulate new data, conditional on model parameters
  y.new[i] ~ dbin(mu.p[i], n[i])
  # pearson chi-square discrepancy for a binomial
  # e is small value to avoid division by zero
  chi2b.data[i] <- ((y[i] - mu.p[i] * n[i]) /
    sqrt((mu.p[i] + e) * n[i] * (1 - mu.p[i] - e)))^2
  chi2b.sim[i] <- ((y.new[i] - mu.p[i] * n[i]) /
    sqrt((mu.p[i] + e) * n[i] * (1 - mu.p[i] - e)))^2
  # freeman-tukey discrepancy for a binomial
  ftd.data[i] \leftarrow (sqrt(y[i]) - sqrt(p[i] * z[i] * n[i]))^2
  ftd.sim[i] \leftarrow (sqrt(y.new[i]) - sqrt(p[i] * z[i] * n[i]))^2
}
# bayesian p-value for chi-square discrepancy
```

```
d.chi2b.data <- sum(chi2b.data)
d.chi2b.sim <- sum(chi2b.sim)
p.chi2b <- step(d.chi2b.sim - d.chi2b.data)

# bayesian p-value for freeman-tukey discrepancy
d.ftd.data <- sum(ftd.data)
d.ftd.sim <- sum(ftd.sim)
p.ftd <- step(d.ftd.sim - d.ftd.data)
}</pre>
```

61 References

- MacKenzie, D. I., Nichols, J. D., Lachman, G. B., Droege, S., Andrew Royle, J., & Langtimm,
- ⁶³ C. A. (2002). Estimating site occupancy rates when detection probabilities are less than
- one. Ecology, 83(8), 2248-2255.